

# Everyday Uses for ODS

Author: Philip R Holland, Consultant,  
Holland Numerics Ltd

## Abstract

This paper describes a number of uses for ODS HTML, ODS CSVALL and ODS DOCUMENT statements for reporting.

## Disguising a web page

### Code

Read as Excel spreadsheet by Microsoft Excel and StarOffice Calc:

```
ODS HTML FILE="report.xls";
```

Read as Word document by Microsoft Word and StarOffice Writer:

```
ODS HTML FILE="report.doc";
```

Read as PowerPoint presentation by Microsoft PowerPoint and StarOffice Impress.

```
ODS HTML FILE="report.ppt";
```

### Uses

- Generating formatted Microsoft Office documents directly from SAS code without any requirement to license SAS/ACCESS component.

### Drawbacks

- Underlying data is stored in an HTML web page, so data values are displayed according to the default actions in the program reading the file, e.g. numeric values with leading zeroes are re-formatted within Excel and Word, the displayed precision of numeric values may be reduced in Excel, numeric and character values in the same column are right and left justified respectively within Word.

## Creating reports in parallel

### Code

Reports containing the same or overlapping parts of an overall report can be generated using indexed ODS statements:

```
ODS HTML(1) FILE="report1.xls";  
ODS RTF FILE="report1.rtf";
```

```
PROC REPORT DATA=summary NOWD;  
.....  
RUN;  
  
ODS HTML(2) FILE="report2.doc";  
ODS HTML(3) FILE="report3.xls";  
  
PROC REPORT DATA=details NOWD;  
.....  
RUN;  
  
ODS HTML(3) CLOSE;  
  
PROC REPORT DATA=exceptions NOWD;  
.....  
RUN;  
  
ODS HTML(2) CLOSE;  
ODS RTF CLOSE;  
ODS HTML(1) CLOSE;
```

The **summary**, **details** and **exceptions** reports will appear in **report1.xls** and **report1.rtf**, the **details** and **exceptions** reports will appear in **report2.doc**, and the **details** report will appear in **report3.xls**.

### Uses

- Multiple reports containing overlapping tables can be generated using a single SAS program.

### Drawbacks

- Reports can only contain information from adjacent SAS steps.

## Saving your reports for later

The code below is flagged as “Experimental” in the SAS 8.2 Log, but will still run. It has “Production” status in SAS 9.1.3.

### Code

Saving your reports to a SAS Item Store.

```
ODS DOCUMENT NAME=work.doc1;  
PROC MEANS DATA=sashelp.class;  
  CLASS age;  
  VAR height weight;  
RUN;  
  
PROC TABULATE DATA=sashelp.class;  
  CLASS age;  
  VAR height weight;  
  TABLE (age ALL)  
         , (height*MEAN weight*MEAN);  
RUN;  
  
ODS DOCUMENT CLOSE;
```

Copying your Item Store file.

```
PROC COPY IN=work OUT=sasuser;  
  SELECT doc1 (MEMTYPE=ITEMSTOR);  
RUN;
```

Producing reports from an Item Store without having to rerun your report processing.

```
ODS HTML FILE="c:\temp\document.htm"
      STYLE=Default;

PROC DOCUMENT;
  DOC;
  DOC NAME=work.doc1;
  LIST/LEVELS=ALL;
  REPLAY;
RUN;
QUIT;

ODS HTML CLOSE;
```

### Uses

- All the reports generated by a program run can be stored, prior to printing, in a single file. This file can then be used to generate reports selectively to 1 or more destinations without the need to rerun the original data processing.
- SAS Item Store files have the “sas7bitm” suffix in Windows Explorer and UNIX File Manager.

### Drawbacks

- By default Item Store members cannot be seen in the SAS Explorer window. To allow them to be seen in SAS Explorer in SAS 8.2, carry out the following changes to the Explorer options:
  - (a) Start the SAS Explorer with the **EXPLORER** command, or select **Tools => Options => Explorer**.
  - (b) Select the **Members** tab.
  - (c) Select **ITEMSTOR** in the **Type** list. You may have to check the **View Undefined Types** box first.
  - (d) Click the **Edit** button.
  - (e) Click the **Select Icon** button, select the **itemstor** icon in the **Files, Folders, and Reports** group, and click the **OK** button.
  - (f) Click the **OK** button.
  - (g) Click the **Close** button.
- To allow them to be seen in SAS Explorer in SAS 9.1.3, carry out the following changes to the Explorer options:
  - (a) Start the SAS Explorer with the **EXPLORER** command, or select **Tools =>**

### Options => Explorer.

- (b) Select the **Members** tab.
  - (c) Select **ITEMSTOR** in the **Type** list.
  - (d) Click the **Unhide** button, so that it changes to **Hide**. A **Select Icon** windows will appear, select the **itemstor** icon in the **Files, Folders, and Reports** group, and click the **OK** button.
  - (e) Click the **OK** button.
- Currently certain SAS procedures, e.g. **PROC PRINT** and **PROC REPORT**, do not support the DOCUMENT destination.
  - Other SAS procedures, e.g. **PROC FREQ**, can write to the DOCUMENT destination, but reports generated from that Item Store do not look the same as when the reports are sent directly to the final ODS destination. **PROC FREQ** reports from Item Stores look similar to **PROC TABULATE** reports.

### Contact details

The author is a consultant for Holland Numerics Ltd and can be contacted at the following address:

Philip R Holland  
address: Holland Numerics Ltd  
94 Green Drift  
Royston  
Herts. SG8 5BT  
UK  
e-mail: phil@hollandnumerics.com  
web: www.hollandnumerics.com  
tel. (mobile): +44-(0)7714-279085

*SAS and all other SAS Institute Inc. product or service names are registered trademarks or trademarks of SAS Institute Inc. in the USA and other countries. ® indicates USA registration.*